<commit_before>! Copyright (C) 2008, 2009 <NAME>, <NAME>.
! See http://factorcode.org/license.txt for BSD license.
USING: accessors arrays calendar combinators.short-circuit fry
continuations debugger io.directories io.files io.launcher
io.pathnames io.encodings.ascii kernel make mason.common mason.config
mason.platform mason.report mason.notify namespaces sequences
quotations macros ;
IN: mason.child

: make-cmd ( -- args )
    gnu-make platform 2array ;

: make-vm ( -- )
    "factor" [
        <process>
            make-cmd >>command
            "../compile-log" >>stdout
            +stdout+ >>stderr
        try-process
    ] with-directory ;

: builds-factor-image ( -- img )
    builds/factor boot-image-name append-path ;

: copy-image ( -- )
    builds-factor-image "." copy-file-into
    builds-factor-image "factor" copy-file-into ;

: factor-vm ( -- string )
    target-os get "winnt" = "./factor.com" "./factor" ? ;

: boot-cmd ( -- cmd )
    factor-vm
    "-i=" boot-image-name append
    "-no-user-init"
    3array ;<commit_msg>mason.child: bootstrap with SSE3 on x86<commit_after>! Copyright (C) 2008, 2009 <NAME>, <NAME>.
! See http://factorcode.org/license.txt for BSD license.
USING: accessors arrays calendar combinators.short-circuit fry
continuations debugger io.directories io.files io.launcher
io.pathnames io.encodings.ascii kernel make mason.common mason.config
mason.platform mason.report mason.notify namespaces sequences
quotations macros ;
IN: mason.child

: make-cmd ( -- args )
    gnu-make platform 2array ;

: make-vm ( -- )
    "factor" [
        <process>
            make-cmd >>command
            "../compile-log" >>stdout
            +stdout+ >>stderr
        try-process
    ] with-directory ;

: builds-factor-image ( -- img )
    builds/factor boot-image-name append-path ;

: copy-image ( -- )
    builds-factor-image "." copy-file-into
    builds-factor-image "factor" copy-file-into ;

: factor-vm ( -- string )
    target-os get "winnt" = "./factor.com" "./factor" ? ;

: boot-cmd ( -- cmd )
    [
        factor-vm ,
        "-i=" boot-image-name append ,
        "-no-user-init" ,
        target-cpu get { "x86.32" "x86.64" } member? [ "-sse-version=30" , ] when
    ] { } make ;
{'fraction_non_alphanumeric': 0.1447132616487455, 'fraction_numerical': 0.014336917562724014, 'mean_word_length': 2.4945226917057903, 'pattern_counts': {'":': 0, '<': 9, '<?xml version=': 0, '>': 21, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1758774', 'n_tokens_mistral': 754, 'n_tokens_neox': 722, 'n_words': 172}